
Groovy 
2 

 a new(ish) programming language for the JVM 

 an agile, dynamic programming language like 

Python, PERL and Ruby 

 completely interoperable with conventional Java 

 makes life fun again! 



The Question 
3 

Wrong Question 

Better Question 



The Answer 
4 

They ‘complete’ each other 



The Answer… 

 all these companies, products and users are benefitting 
from the Gr8 technologies 

 Spring, Seam, JBoss, IntelliJ, Eclipse, JDeveloper/ADF, 
SoapUI, Selenium, Jenkins, Freemind, Confluence, 
OpenOffice… 

 eHarmony, European Patent Office, Wired.com, Vodafone, 
Netflix, Suncorp, Mincom, Atlassian, Thoughtworks, Canoo,… 

 Sky.com 

 ~1 billion monthly page impressions 

 Barack Obama! 

 me! 

 



The Gr8 Technologies 

 a complete, powerful ecosystem 

 Grails 

 Griffon 

 Gant 

 Gradle 

 GPars 

 Geb 

 Betamax 

 Spock 

 …many, many more 



7 http://www.asert.com/pubs/Groovy/Groovy.pdf 

16 Many thanks to maestro Paul! 
Many thanks to maestro Paul! http://www.asert.com.au 



Aims 
17 

 put the FUN back into work! 

 simplify developers lives 

 convention-over-configuration 

 become more ‘agile’ 

 make better tools 

 scripting 

 builders and slurpers 

 make building tools easier 

 Domain-Specific Languages 



Scripting 
18 

 no more need for shell scripts, PERL, etc. 

final DIR = /C:\Users\Bob Brown\Desktop/ 

 

datPagesScanner = new AntBuilder().fileScanner { 

    fileset(dir: DIR, includes: '*.dat') 

} 

 

new File("${DIR}/copy.txt").withWriter { file -> 

  datPagesScanner.each { datFile -> 

    datFile.eachLine { line -> 

    if (line =~ /^[AEOIUaeiou].*/) 

      file.writeLine(line) 

    } 

  } 

} 



Builders 

 simplify creation of HTML, XML, JSON, Swing UI, … 

builders come into their own within Grails... 

import groovy.xml.MarkupBuilder 

 

def builder = new MarkupBuilder (); 

 builder.html { 

    head { 

      title "This is Marked-Up HTML" 

      style type:'text/css', ".emph { background: gray }" 

    } 

    body { 

      p 'class':'emph', "This uses Groovy's MarkupBuilder" 

      p(/Good, isn't it!/) 

    } 

 } 

 



Slurpers 

 just compare… 

import java.io.*; 

import javax.xml.parsers.*; 

import org.w3c.dom.*; 

 

public class XMLReader { 

  public static void main(String argv[]) throws Exception { 

    File file = new File("items.xml"); 

    DocumentBuilderFactory dbf = DocumentBuilderFactory.newInstance(); 

    DocumentBuilder db = dbf.newDocumentBuilder(); 

    Document doc = db.parse(file); 

    doc.getDocumentElement().normalize(); 

    NodeList nodeLst = doc.getElementsByTagName("an-item"); 

    for (int s = 0; s < nodeLst.getLength(); s++) { 

      Element anItem = (Element) nodeLst.item(s); 

      System.out.println(anItem.getAttribute("the-id") + ": " + 

                anItem.getChildNodes().item(0).getNodeValue()); 

    } 

  } 

} 

“Nothing Makes You Want Groovy More Than XML...” 
—http://kousenit.wordpress.com/2008/03/12/nothing-makes-you-want-groovy-more-than-xml/ 

<?xml version="1.0" encoding="UTF-8"?> 

<items> 

  <an-item the-id="0">This is item 0</an-item> 

  […elided…] 

</items> 



Slurpers… 
21 

 easily consume structured data 

 

 

 

 all sorts of slupers 

 Config, XML, JSON, CSV 

items = new XmlSlurper().parse(new File('items.xml')) 

 

items?.'an-item'.each { 

  println "${it.'@the-id'.text()}: ${it.text()}" 

} 



Grapes 

22 

 Groovy Advanced Packaging Engine 

 lets you grab a script’s needed resources 



Grapes… 
23 

@Grapes( 

  @Grab('com.googlecode.gbench:gbench:0.4.1-groovy-2.1') 

) 

 

def r = benchmark(verbose: true) { 

    'Each' { 

        def i = 0 

        (1..100000000).each { i ++ } 

    } 

    'For' { 

        def i = 0 

        for (x in 1..100000000) { i ++ } 

    } 

} 

r.prettyPrint() 

Environment 

=========== 

* Groovy: 2.1.2 

* JVM: Java HotSpot(TM) 64-Bit Server VM (23.7-b01, Oracle Corporation) 

    * JRE: 1.7.0_17 

    * Total Memory: 127.875 MB 

    * Maximum Memory: 127.875 MB 

* OS: Windows 8 (6.2, amd64)  

 

Options 

======= 

* Warm Up: Auto (- 60 sec) 

* CPU Time Measurement: On 

 

Warming up "Each"... 

Measuring "Each"... 

Warming up "For"... 

Measuring "For"... 

            user  system         cpu        real 

 

Each  7859375000       0  7859375000  8111961507 

For   2250000000       0  2250000000  2304534377 



Better JDBC With GroovySQL 

Domain Specific Languages 

 ‘little languages’ for well-defined purposes 
presentation('Gr8 Technologies') { 

  used 'laptop-imp' duration 1.2.hours 

  printed 5.pages on 'hp-printer' 

  presented 1.hour date '9/5/2013' at 'HKJUG' 

} 



Domain Specific Languages… 

 very simple to define! 

 command chain expressions 
[…elided…] 

 

def presented(hours) { 

   ['date': { date -> 

     ['at': {  where -> 

       // may want something more interesting…maybe insert into a db/send an email… 

       println "presented $hours hour(s) on $date at $where" 

     }] 

   }] 

} 

 

def used(equipment) { 

  ['duration': { dur ->  

    println "used $equipment for $dur hour(s)" 

  }] 

} 

 

def printed(pages) { 

  ['on': { equipment -> 

    println "$pages page(s) were printed on '$equipment'" 

  }] 

} 



GPars 
27 

 parallel programing made easy(er) 



Interoperability 
28 

 use any old Java library 

 JFugue 

 pure Java midi/music framework 

 no problem to use with Groovy 

import org.jfugue.* 

 

final darthVaderTheme = new Pattern( 

""" 

T160 I[Cello]  

G3h G3h G3h Eb3q Bb3i G3qi Eb3q Bb3i G3hi 

Rq 

D4h D4h D4h Eb4q B3i G3qi Eb3q B3q G3h 

""") 

 

new Player().with { player ->  

  player.play(darthVaderTheme) 

  player.saveMidi(darthVaderTheme, 

                  new File(/C:\DEVELOPMENT\Gradle\StarWars\darth.mid/)) 

} 



WSLite 

 no-frills SOAP and REST webservice clients 

 Simple & powerful 

@Grab(group='com.github.groovy-wslite', module='groovy-wslite', version='0.1') 

import wslite.soap.* 

 

def soapClient = new SOAPClient("http://www.webservicex.net/WeatherForecast.asmx") 

def response = soapClient.send { 

    version SOAPVersion.V1_2 

    body { 

        GetWeatherByZipCode(xmlns:"http://www.webservicex.net") { 

            ZipCode("93657") 

        } 

    } 

} 

 

assert "SANGER" == 

  response.GetWeatherByZipCodeResponse.GetWeatherByZipCodeResult.PlaceName.text() 

assert 200 == response.http.statusCode 

assert "OK" == response.http.statusMessage 

assert "ASP.NET" == response.http.headers["X-Powered-By"] 



Performance 

 Big Topic! 

 dynamic language implies some overhead 

 but often within a few % of ‘pure’ Java 

 developer vs program 

 Java 7’s new invokeDynamic opcode 

 AST Transforms 

 @Typechecked, @CompileStatic 

 lots of ongoing work 



Performance… 

 @CompileStatic 

 > 2 * speedup 

http://www.sergeydolgopolov.me/2012/07/groovy-20-has-been-released-testing-new.html 

http://code.google.com/p/caliper/wiki/JavaMicrobenchmarks 



Gant 
32 

 scripting Ant tasks using Groovy 

 no XML!  

includeTargets << gant.targets.Clean 

cleanPattern << ['**/*.class', '**/*~', '**/*.bak', '**/*.OLD'] 

cleanDirectory << 'build' 

 

taskdef (name: 'groovyc', classname: 'org.codehaus.groovy.ant.Groovyc') 

 

ant.path(id: 'runtimeClasspath') { 

  pathelement(location: 'build') 

  pathelement(location: 'C:/DEVTOOLS/gant-1.8.1/lib/groovy-all-1.6.5.jar') 

} 

 

target(name: 'default') { 

  ant.mkdir(dir: 'build') 

  groovyc (srcdir: 'src', destdir: 'build', verbose: false) 

  java(classname: 'HelloWorld', fork:true, dir: 'build',  

       classpathref: 'runtimeClasspath') { 

    arg(line: 'FRED') 

  } 

} 



Gradle 
33 

 more convention, less configuration 

 no more “{classpath, xml, maven} hell” 

http://prezi.com/ypobbs_ikvtj/gradle-a-better-way-to-build/ 



Vert.X 
35 

 “Effortless asynchronous application development 

for the modern web and enterprise” 

 in the style of Node.js… 

 …but polyglot 

 Groovy/Java, JavaScript, CoffeeScript, Ruby, Python 

 …and on the JVM 

 
vertx.createHttpServer().requestHandler { req -> 

    def file = req.uri == "/" ? "index.html" : req.uri 

    req.response.sendFile "webroot/$file" 

}.listen(8080) 



Grails 
37 

“Grails is a dynamic web application framework built on 

Java and Groovy, leveraging best of breed APIs 

including Spring, Hibernate and SiteMesh. Grails brings 

to Java and Groovy developers the joys of convention-

based rapid development while allowing them to 

leverage their existing knowledge and capitalize on the 

proven and performant APIs Java developers have been 

using for years.” 



Grails... 

 a full CRUD HTML5 webapp  

 minimal effort 

// persistent domain class 

class Cat { 

  String name 

  short age 

  String disposition 

} 

// controller class 

class CatController { 

    static scaffold = true 

} 



Griffon 

 grails-like rich Swing client framework  

 standardised build system ‘inspired’ by Grails 

 ‘...by “inspired” I mean “taking large chunks of Grails code to 
bootstrap the codebase...”’ 

 a structure that supports/rewards MVC 

 and enables easy thread-handling 
 one of the biggest hurdles for Swing developers 

 Groovy goodness: builders, @Bindable annotation, metaclass 
method injection, scripts, etc.  

 declarative layout of GUI code in the view 

 plugins  

 automatic packaging and signing for WebStart, Applet, and 
traditional application deployment 

 from the SAME source 
“Griffon is to the desktop what 

Grails is to the web.” 
—http://groovy.dzone.com/news/hello-griffon 

http://griffon.codehaus.org/


Griffon… 

 twittersphere 

 created as a technology demonstration for JavaOne 2009 

 won the Script Bowl 

 against Jython, Clojure, 
Scala and JRuby 

 mashup with NASA  
World Wind 

 locates twitterers on  
an animated  
world map 

 in real-time! 

 only 681 LOC! 



Griffon... 

import java.awt.event.ActionEvent 

 

class GRIController { 

    def model 

    def view 

 

  def executeScript(ActionEvent evt = null) { 

    doOutside {             

          model.greeting = 'HI!' 

    } 

  } 

} 

import groovy.beans.Bindable 

 

class GRIModel { 

  @Bindable def greeting = "" 

} 

application(title:'GRI',   

            pack:true, 

            locationByPlatform:true) { 

  borderLayout() 

  hbox(constraints:NORTH) { 

    button("Execute", actionPerformed:controller.&executeScript) 

  } 

  hbox(constraints:SOUTH) { 

    hstrut(5) 

    label("Result:") 

    hstrut(5) 

    label(text:bind {model.greeting}) 

  } 

} 



Testing 

 dynamic languages don’t have the help of a strong 

type system 

 typos, etc. not uncovered until run-time* 

 testing required 

 more? 

 but testing is always required  

so not a problem? 

* but good IDEs can help quite a lot…many errors can be surfaced at edit-time 



Testing… 

class Grader { 

    def expectedAnswers 

    def graderFileReader 

 

    def grade(String s) { 

        def candidateAnswers = graderFileReader.readGradesListFromFile(s) 

        grade(candidateAnswers) 

    } 

 

    def grade(List candidateAnswers) { 

        if (expectedAnswers?.size() != candidateAnswers?.size()) 

            -1.0 

        else { 

            def count = 0 

            expectedAnswers.eachWithIndex {o,index -> 

                if (o == candidateAnswers[index]) count ++ 

            }   

            count / expectedAnswers.size() 

        } 

    } 

} 

class GraderFileReader { 

    def readGradesListFromFile(name) { 

        def f = new File(name) 

        if (!f.exists()) 

          throw new Exception("File $name does not exist.") 

        def txt = f.text 

        txt?.split(',') as List 

    } 

} 



Spock 

 unit testing framework based on specifications 

 “given – when– then” stories 



Spock… 
46 

public class GraderSpecification extends Specification { 

    def grader 

 

    def "The perfect paper"() { 

        when: "A perfect answer is presented" 

          def result = grader.grade(['a','b','c']) 

        then: "The grade should be 100%" 

          result == 1.0 

    } 

 

    def "The worst paper"() { 

        when: "No answers are given" 

          def result = grader.grade([]) 

        then: "An error should be indicated" 

          result == -1.0 

    } 

 

    def "A poor paper"() { 

        when: "A fairly poor paper is presented" 

          def result = grader.grade(['a','c','b']) 

        then: "The grade should be 33%" 

          result closeTo(0.33D, 0.01D) 

    } 

 

    def setup()  { grader = new Grader(expectedAnswers: ['a','b','c']) } 

    def cleanup() { grader = null } 

} 



Spock 
47 



Spock… 

 table-driven parameterised testing 

public class GraderSpecification2 extends Specification { 

    @AutoCleanup(quiet = true) 

    def grader = new Grader(expectedAnswers: ['a', 'b', 'c']) 

 

    @Unroll("(Inline) #iterationCount: grade for #paper is #res") 

    def "Grader with papers given inline"() { 

        expect: "Grade an individual paper" 

            that grader.grade(paper), closeTo(res, 0.01D) 

 

        where: "With the following papers" 

            paper                | res 

            ['a', 'b', 'c']      | 1.0D 

            ['a', 'b', 'd']      | 0.66D 

            ['a', 'c', 'b']      | 0.33D 

            ['x', 'y', 'z']      | 0.0D 

            ['c', 'a', 'b']      | 0.0D 

            ['a', 'b']           | -1.0D 

            ['a', 'b', 'c', 'd'] | -1.0D 

            []                   | -1.0D 

            null                 | -1.0D 

    } 

} 

“Green is Good” 



Spock… 

 mocking and expectations 
class GraderSpecification3 extends Specification { 

    @AutoCleanup(quiet = true) 

    def grader = new Grader(expectedAnswers: ['a','b','c']) 

 

    def "Given a mock file"() { 

        setup: "Establish the grader with a mocked GraderFileReader" 

          def graderFileReader = Mock(GraderFileReader) 

          grader.graderFileReader = graderFileReader 

          1 * graderFileReader.readGradesListFromFile(_) >> ['a','b','c'] 

          0 * _._ 

 

        when: "Grade a paper's answers from a given file" 

          def res = grader.grade('rsrc/100pct.txt') 

         

        then: "Ensure expected behaviour" 

          res == 1.0D 

    } 

} 



Geb 

 functional testing for the web 

 easy-to-use DSL 

 no nasty C or XML like competing tools 



Geb… 

@Grapes([ 

    @Grab("org.gebish:geb-core:0.9.0"), 

    @Grab("org.seleniumhq.selenium:selenium-firefox-driver:2.26.0"), 

    @Grab("org.seleniumhq.selenium:selenium-support:2.26.0") 

]) 

import geb.* 

    

Browser.drive { 

  go "http://www.google.com/" 

  assert title == "Google" 

  

  $("input", name: "q").value("wikipedia") 

  $("input", value: "Google Search").click() 

  

  assert title.endsWith("Google Search") 

  

  def firstResultLink = $("li.g", 0).find("a.l") 

  assert firstResultLink.text() == "Wikipedia, the free encyclopedia" 

} 



Betamax 

 test proxy/framework 

 first time: record; then: replay 

 breaks dependencies between teams/systems 

during test/development 

 functional mocking 

 regression testing 



Betamax… 

import geb.spock.GebSpec 

import betamax.* 

import org.junit.* 

import spock.lang.* 

 

class TransentiaSpec extends GebSpec { 

    @Rule recorder = new Recorder() 

 

    @Betamax(tape="transentia.betamax.tape") 

    def "go to Transentia home page"() { 

        setup: 

            browser.driver.setProxy("localhost", 5555) 

        when: 

            go "http://www.transentia.com.au/" 

        then: 

            title.startsWith('Transentia') 

        and: 

          // some basic content checks 

          def about = $("div.about") 

          def aboutTitle = about.find("h2.title") 

          aboutTitle.text() == "About Transentia" 

          aboutTitle.next().text().contains("Gr8") 

    } 

} 



CodeNarc 

 code inspections 

 configurable command-line tool 

 for use with Jenkins/development teams 

 checking for common whoopsies, gotchas, etc. 

 inconsistencies, unneeded/dead code 

 checks subtle/uncommon issues 

 threading, memory, resource usage 



CodeNarc… 

ruleset { 

    description 'A Sample Groovy RuleSet' 

    AssignmentInConditional 

    StaticCalendarField 

    SynchronizedOnBoxedPrimitive 

    ReturnsNullInsteadOfEmptyCollection 

    SimpleDateFormatMissingLocale 

    DuplicateNumberLiteral 

    CatchIllegalMonitorStateException 

    … 

} 



Cobertura 

 code coverage testing 

 command-line tool 

 configurable 

 show what has been tested 

 guide what further tests need to be created 



Cobertura 
57 



Summary 
60 

• an agile and dynamic language for the Java Virtual Machine  

• builds upon the strengths of Java but has additional power features inspired by 

languages like Python, Ruby and Smalltalk  

• makes modern programming features available to Java developers with almost-

zero learning curve  

• supports Domain-Specific Languages and other compact syntax so your code 

becomes easy to read and maintain  

• makes writing shell and build scripts easy with its powerful processing primitives, 

OO abilities and an Ant DSL  

• increases developer productivity by reducing scaffolding code when developing 

web, GUI, database or console applications  

• simplifies testing by supporting unit testing and mocking out-of-the-box  

• seamlessly integrates with all existing Java objects and libraries  

• compiles straight to Java bytecode so you can use it anywhere you can use Java  



Summary 
61 

 …of the summary 



Learn More 

 Resources 

 http://www.transentia.com.au 

 user@groovy.codehaus.org 

 http://groovy.codehaus.org 

 http://gradle.org 

 http://griffon.codehaus.org 

 http://grails.codehaus.org 

 http://jenkins-ci.org 

 http://gant.codehaus.org 

 http://gmetrics.sourceforge.net 

 http://cobertura.sourceforge.net 

 http://easyb.org 

 http://jfugue.org 

 http://jscience.org 

 http://codenarc.sourceforge.net 

 http://code.google.com/p/spock 

 http://robfletcher.github.com/betamax 

 http://gebish.org 

 http://mrhaki.com 

 http://groovyblogs.org 

 http://groovymag.com 

 http://vertx.io/ 

END 

謝謝您們的聆聽 

(questions?) 

Thanks for listening 


